Supplemental Material for Online Publication

From attentional lapses to intentional lapses? Monitoring behavior and intraindividual variability in time-based prospective memory

Joly-Burra, E., Hass, M., Laera, G., Ghisletta, P. Kliegel, M., & Zuber, S.

The following supplemental material is divided into two appendices. Appendix A reports Mplus syntaxes to run all dynamic structural equation modeling (DSEM) models reported in the manuscript, while Appendix B details corresponding prediction equations. Appendix C contains estimates for Model OT and overall PM costs to iM, and iSD. De-identified data are openly available at https://osf.io/53ucf/. The E-Prime2 script for the time-based PM task is openly available online at https://cigev.unige.ch/openscience/.

APPENDIX A: Mplus Syntaxes

MODEL 0

```
  TITLE:  Model 0 ! Computed to estimate the random effect at the between level (i.e., factor scores) of
the within-person effects, and their respective correlations.


  DATA:   FILE = Data_IIV_monitoring_TBPM.txt ; ! Specifies data file

    VARIABLE:
  NAMES = ! Specifies variable name for each column in the database
  Subject Trial Trial_number PM_window Trial_ACC All_RT RT_raw RT iSD_raw iSD
  PM AbsoluteCC RelativeCC OT Age Sex Education Mill_Hill WAIS_Matrices;

                CLUSTER = Subject; ! specifies that trials are nested within participants
                USEVAR = RT PM_window Trial_number;
                between = ;
                within = PM_window Trial_number; ! within-person level variable
                MISSING = . ; ! missing variables are indicated by a dot
                LAGGED = RT(1); ! the autoregressive parameter is an AR(1) process
                TINTERVAL = Trial(1);  ! time interval for the autoregressive process

  ANALYSIS:   TYPE = TWOLEVEL RANDOM; ! Defines the DSEM estimator properties
              ESTIMATOR = BAYES;
              PROCESSORS = 8;
              fbiter = (50000);
              BSEED = 41;
              THIN = 10;

  MODEL:
    %WITHIN%
     phi | RT ON RT&1; ! Specifies autoregressive parameter phi of lag 1 (time-structured IIV)
     slowing | RT ON PM_window; ! Specifies the slowing parameter
```

```
    trial | RT ON Trial_number; ! Specifies the effect of trial to control for possible trends in OT RT
!through the task (i.e., participants getting generally faster or slower as they progress in the task)

    %BETWEEN%

        RT phi slowing WITH phi slowing; ! Specifies correlations between random effects


  PLOT: TYPE = PLOT2; ! posterior parameter distributions, trace and autocorrelation plots
        TYPE = PLOT3; ! histograms of estimated factor scores, and time series plots
        FACTORS = ALL (100); ! distribution of all factor scores

  OUTPUT:     TECH8 ! optimization history
              STANDARDIZED ! standardized parameter estimates and their standard errors and R-square
              RESIDUAL; ! residuals for the observed variables in the analysis
              FSCOMPARISON; ! comparison of between-level estimated factor scores
```

Model 1

```
TITLE:  Model 1 ! IIV predicts time monitoring behavior


  DATA:  FILE = Data_IIV_monitoring_TBPM.txt;

    VARIABLE:
NAMES =
Subject Trial Trial_number PM_window Trial_ACC All_RT RT_raw RT iSD_raw
iSD PM AbsoluteCC RelativeCC OT Age Sex Education Mill_Hill WAIS_Matrices;

              CLUSTER = Subject;
              USEVAR = RT iSD PM_window AbsoluteCC RelativeCC Age Trial_number;
              between = iSD AbsoluteCC RelativeCC Age;
              within = PM_window;
              MISSING = .;
              LAGGED = RT(1);
              TINTERVAL = Trial(1);

  DEFINE: CENTER iSD AbsoluteCC RelativeCC Age (GRANDMEAN);
  STANDARDIZE iSD AbsoluteCC RelativeCC Age;

  ANALYSIS:   TYPE = TWOLEVEL RANDOM;
              ESTIMATOR = BAYES;
              PROCESSORS = 8;
              fbiter = (50000);
              BSEED = 41;
              THIN = 10;


  MODEL:
   %WITHIN%
     phi | RT ON RT&1;
     slowing | RT ON PM_window;
     trial | RT ON Trial_number;
```

```
 %BETWEEN%

   Slowing ON RT iSD phi Age; ! mean RT (iM), iSD, phi, and age predict slowing
   AbsoluteCC ON RT iSD phi Age; ! the same variables predict absolute clock-checking
   RelativeCC ON RT iSD phi Age; ! the same variables predict relative clock-checking

   RT ON Age; ! age predicts iM
   iSD ON Age; ! age predicts iSD (net IIV)
   phi ON Age; ! age predicts phi (time-structured IIV)

   RT iSD WITH iSD phi; ! allows correlation between the residual variances for iM, iSD and phi
   AbsoluteCC RelativeCC WITH RelativeCC Slowing; ! allows correlation between the residual variances
for all three time monitoring behavior indicators


PLOT: TYPE = PLOT2;
      TYPE = PLOT3;
      FACTORS = ALL (100);

OUTPUT:    TECH8
           STANDARDIZED
           RESIDUAL;
           FSCOMPARISON;
```

Model 2

```
TITLE:  Model 2 ! Only IIV predicts OT and PM accuracies


  DATA:  FILE = Data_IIV_monitoring_TBPM.txt;

    VARIABLE:
NAMES =
Subject Trial Trial_number PM_window Trial_ACC All_RT RT_raw RT iSD_raw
iSD PM AbsoluteCC RelativeCC OT Age Sex Education Mill_Hill WAIS_Matrices;

              CLUSTER = Subject;
              USEVAR = RT iSD PM_window OT PM Age Trial_number;
              between = iSD OT PM Age;
              within = PM_window Trial_number;
              MISSING = . ;
              LAGGED = RT(1);
              TINTERVAL = Trial(1);

  DEFINE: CENTER iSD Age (GRANDMEAN);
  STANDARDIZE iSD Age;

  ANALYSIS:   TYPE = TWOLEVEL RANDOM;
              ESTIMATOR = BAYES;
              PROCESSORS = 8;
              fbiter = (50000);
              BSEED = 41;
              THIN = 10;


  MODEL:
   %WITHIN%
     phi | RT ON RT&1;
     slowing | RT ON PM_window;
     trial | RT ON Trial_number;
```

```
 %BETWEEN%

   PM ON RT iSD phi Age; ! iM, iSD, phi and Age predict PM performance
   OT ON RT iSD phi Age; ! iM, iSD, phi and Age predict OT performance

   PM WITH OT; ! residual variances for PM and OT performance are allowed to correlate

PLOT: TYPE = PLOT2;
      TYPE = PLOT3;
      FACTORS = ALL (100);

OUTPUT:      TECH8
             STANDARDIZED
             RESIDUAL;
             FSCOMPARISON;
```

Model 3

```
TITLE:  Model 3 ! Both IIV and time monitoring predict OT and PM accuracies


  DATA:  FILE = Data_IIV_monitoring_TBPM.txt;

    VARIABLE:
NAMES =
Subject Trial Trial_number PM_window Trial_ACC All_RT RT_raw RT iSD_raw
iSD PM AbsoluteCC RelativeCC OT Age Sex Education Mill_Hill WAIS_Matrices;

              CLUSTER = Subject;
              USEVAR = RT iSD PM_window OT PM Age AbsoluteCC RelativeCC
                       Trial_number;
              between = iSD OT PM Age;
              within = PM_window Trial_number;
              MISSING = . ;
              LAGGED = RT(1);
              TINTERVAL = Trial(1);

  DEFINE: CENTER iSD Age AbsoluteCC RelativeCC (GRANDMEAN;
  STANDARDIZE iSD Age AbsoluteCC RelativeCC;

  ANALYSIS:   TYPE = TWOLEVEL RANDOM;
              ESTIMATOR = BAYES;
              PROCESSORS = 8;
              fbiter = (50000);
              BSEED = 41;
              THIN = 10;

  MODEL:
   %WITHIN%
     phi | RT ON RT&1;
     slowing | RT ON PM_window;
     trial | RT ON Trial_number;
```

```
 %BETWEEN%

    PM ON RT iSD phi Age AbsoluteCC RelativeCC slowing; ! adding the three time monitoring indicators
    OT ON RT iSD phi Age AbsoluteCC RelativeCC slowing; ! adding the three time monitoring indicators

    PM WITH OT;

PLOT: TYPE = PLOT2;
      TYPE = PLOT3;
      FACTORS = ALL (100);

OUTPUT:     TECH8
            STANDARDIZED
            RESIDUAL;
            FSCOMPARISON;
```

MODEL OT only

```
  TITLE:  Model OT only ! Computed to obtain the random effects at the between level (i.e., factor scores)
for iM and ϕ in the OT only block and thus compute global costs to perform the PM task on top of the OT
(see Appendix C of the present supplemental material)


  DATA:   FILE = Data_IIV_OTOnly.txt;

    VARIABLE:
NAMES = Subject Trial_OTOnly Trial_number_OTOnly Trial_ACC_OTOnly All_RT_OTOnly
RT_raw_OTOnly RT_OT OT_OTOnly iSD_raw_OTOnly iSD_OTOnly Age;
                CLUSTER = Subject;
                USEVAR = RT_OTOnly Trial_number;
                between = ;
                within = Trial_number;
                MISSING = . ;
                LAGGED = RT_OTOnly(1);
                TINTERVAL = Trial_OTOnly (1);

  ANALYSIS:   TYPE = TWOLEVEL RANDOM;
              ESTIMATOR = BAYES;
              PROCESSORS = 8;
              fbiter = (50000);
              BSEED = 41;
              THIN = 10;

  MODEL:
    %WITHIN%
     Phi_OT | RT_OTOnly ON RT_OTOnly&1;
     trial_OT | RT_OTOnly ON Trial_number_OTOnly;

    %BETWEEN%

     RT_OTOnly phi_OT trial_OT WITH phi_OT trial_OT;
```

```
PLOT: TYPE = PLOT2;
      TYPE = PLOT3;
      FACTORS = ALL (100);

OUTPUT:     TECH8
            STANDARDIZED
            RESIDUAL;
            FSCOMPARISON;
```

APPENDIX B: Equation predictions for dynamic structural equation modeling (DSEM)

Following Hamaker and colleagues (Hamaker et al., 2017; 2018), Model 0 first decomposes RT into *within* and *between-person* components as follows:

$$RT_{it} = iM_i + RT_{it}^*, \quad (1)$$

where iM_i is the time-invariant (*between-person*) mean RT for individual *i* while RT_{it}^* represents the (*within-person*) individual deviations from iM_i at trial *t*.

The within-person component RT_{it}^* is decomposed as follows:

$$RT_{it}^* = trial_i Trialnumber_{it} + \phi_i RT_{i,t-1}^* + slowing_i PMwindow_{it} + \zeta_{it}, \quad (2)$$

Where $trial_i$ is the linear effect of trial order to control for possible individual-specific trends in RT through the task, ϕ_i is the first-order autoregressive parameter of individual *i* for RT for two successive trials, $slowing_i$ is the mean slowing of RT of individual *i* for trials located within the PM response window (when $PMwindow_{it}$ = 1), and ζ_{it} is the residual variations in RT at trial *t* not explained by the previous three predictors. Residuals are supposed normally distributed around zero with constant variance σ_ζ^2. For trials located outside of the PM response window, $slowing_i PMwindow_{it}$ = 0, and $RT_{it}^* = trial_i Trialnumber_{it} + \phi_i RT_{i,t-1}^* + \zeta_{it}$. The overall mean RT iM_i, the effect of trend $trial_i$, the autoregressive parameter ϕ_i, and the $slowing_i$ parameter are allowed to vary across persons (hence the subscript *i*). That is, they have random effects (υ) as in:

$$\begin{aligned} iM_i &= \gamma_{iM} + \upsilon_{iM_i}, \\ trial_i &= \gamma_{trial} + \upsilon_{trial_i}, \\ \phi_i &= \gamma_\phi + \upsilon_{\phi_i}, \\ slowing_i &= \gamma_{slowing} + \upsilon_{slowing_i}, \end{aligned} \quad (3)$$

where υ_{iM_i}, υ_{trial_i}, υ_{ϕ_i}, and $\upsilon_{slowing_i}$ are normally distributed, have constant variance σ^2_{iM}, σ^2_{trial}, σ^2_{ϕ}, and $\sigma^2_{slowing}$ respectively, and are allowed to covary ($\sigma_{iM,trial}$, $\sigma_{iM,\phi}$, $\sigma_{iM,slowing}$, $\sigma_{trial,\phi}$, $\sigma_{trial,slowing}$, $\sigma_{\phi,slowing}$) with each other.

As is customary in multilevel modeling, we can combine Equations (1) to (3) to obtain:

$$RT_{it} = \gamma_{iM} + \gamma_{trial} Trialnumber_{it} + \gamma_{\phi} RT^{*}_{i,t-1} + \gamma_{slowing} PMwindow_{t} + \upsilon_{iM_i} + \upsilon_{trial_i} + \upsilon_{\phi_i} + \upsilon_{slowing_i} + \zeta_{it}, \quad (4)$$

where γ_{iM}, γ_{trial}, γ_{ϕ}, and $\gamma_{slowing}$ are the fixed effects of the mean RT (i.e., mean RT averaged across participants), the trend (i.e., linear effect of trial order), the first-order autoregressive parameters (i.e., the mean autoregressive parameter on the whole sample), and the overall slowing of RT during the PM response window, respectively. In turn, parameters υ_{iM_i}, υ_{trial_i}, υ_{ϕ_i}, and $\upsilon_{slowing_i}$ indicate the random effects of mean RT level, trend in RT, first-order autoregressive parameter, and slowing parameter, respectively (between-person variations in mean RT, trend, autoregressive parameter at lag 1, and slowing, respectively).

To investigate the relationship between both aspects of IIV (net and time-structured IIV), time monitoring behavior, and task performance, Models 1, 2, and 3, made full use of the strengths of DSEM by further including between-level covariates (i.e., iSD_i, absolute CC, relative CC, OT, PM, and age).

In Model 1, the overall trait-like mean iM_i, ϕ_i, and iSD_i predicted absolute clock-checking, relative clock-checking and slowing. Hence, between-level outcome variables can be regressed on fixed effects as follows:

$$\begin{aligned} absoluteCC_i &= \beta_{0ac} + \beta_{iMac} iM_i + \beta_{\phi ac}\phi_i + \beta_{iSDac} iSD_i + \beta_{ageac} age_i + \upsilon_{eiac}, \\ relativeCC_i &= \beta_{0rc} + \beta_{iMrc} iM_i + \beta_{\phi rc}\phi_i + \beta_{iSDrc} iSD_i + \beta_{agerc} age_i + \upsilon_{eirc}, \\ slowing_i &= \beta_{0s} + \beta_{iMs} iM_i + \beta_{\phi s}\phi_i + \beta_{iSDs} iSD_i + \beta_{ages} age_i + \upsilon_{eis}, \end{aligned} \tag{5}$$

β_{0ac}, β_{0rc}, and β_{0s} are the intercepts of $absoluteCC_i$, $relativeCC_i$, and $slowing_i$, respectively. β_{iMac}, β_{iMrc}, and β_{iMs} are the regression weights for iM_i. $\beta_{\phi ac}$, $\beta_{\phi rc}$, and $\beta_{\phi s}$ are the respective regression weights for ϕ_i (estimated in equation 3). β_{iSDac}, β_{iSDrc}, and β_{iSDs} are the respective regression weights for iSD_i. β_{ageac}, β_{agerc}, and β_{ages} are the respective regression weights for age_i. Finally, υ_{eiac}, υ_{eirc}, and υ_{eis} are the prediction residuals for $absoluteCC_i$, $relativeCC_i$, and $slowing_i$, respectively, and were allowed to correlate. Given that iM_i, ϕ_i, and iSD_i are regressed on age_i, the corresponding prediction residuals υ_{eiiM}, υ_{eiiSD}, and $\upsilon_{ei\phi}$ were allowed to correlate.

In Model 2, age_i, iM_i, ϕ_i, and iSD_i predicted both prospective memory (PM) and ongoing task (OT) performance. Finally, in Model 3, $absoluteCC_i$, $relativeCC_i$, and $slowing_i$ were further included as predictors of PM and OT. Following the same logic, prediction equations for Models 2 and 3 can be written as follows:

$$\begin{aligned} PM_i &= \beta_{0PM} + \beta_{iMPM} iM_i + \beta_{\phi PM}\phi_i + \beta_{iSDPM} iSD_i + \beta_{ageac} age_i + \upsilon_{eiPM} \\ &(+\beta_{absoluteCCPM} absoluteCC + \beta_{relativeCCPM} relativeCC + \beta_{slowingPM} Slowing_i), \\ OT_i &= \beta_{0OT} + \beta_{iMOT} iM_i + \beta_{\phi OT}\phi_i + \beta_{iSDOT} iSD_i + \beta_{ageOT} age_i + \upsilon_{eiOT} \\ &(+\beta_{absoluteCCOT} absoluteCC + \beta_{relativeCCOT} relativeCC + \beta_{slowingOT} slowing_i), \end{aligned} \tag{6}$$

where predictors in parentheses were only included in Model 3. All predictors were allowed to correlate at the between-level. The prediction residuals for PM and OT performance, υ_{eiPM} and υ_{eiOT}, were also allowed to correlate in both models.

References

Hamaker, E. L., Asparouhov, T., Brose, A., Schmiedek, F., & Muthén, B. (2018). At the Frontiers of Modeling Intensive Longitudinal Data: Dynamic Structural Equation Models for the Affective Measurements from the COGITO Study. *Multivariate Behavioral Research*, 1-22. https://doi.org/10.1080/00273171.2018.1446819

Hamaker, E.; Asparouhov, T.; Muthén, B. O. (2017, March) *Dynamic Structural Equation Modeling of Intensive Longitudinal Data Using Mplus Version 8*. https://www.statmodel.com/download/HamakerDSEMforPSMG.pdf

APPENDIX C: Estimates for Model OT and overall PM costs to iM, and iSD

The intraclass correlation for logged RT nested within individuals in the OT only block was 0.32, meaning that there was more variability in RT within individuals (68%) than between individuals (32%). As a comparison, the intraclass correlation in the PM block was 0.18, indicating that there were more marked interindividual differences in the OT only block than in the PM block.

To establish a baseline in OT performance and variability, we computed an additional OT only block model, in which we estimated iM, ϕ and trial effects at the within-person level. At the between-person level, random effects were allowed to covary (see Model OT only syntax in Appendix A). This model is similar to Model 0 in the PM clock, but without the effect of slowing. We report mean fixed effects, random variances, and their corresponding covariances in both raw and within-level standardized metrics for Model OT only in Table C1 below.

Table C1. Posterior Means [and 95% CIs] of Fixed Effects and Random Effect Variances from Model OT.

Model OT			
	Fixed effects		Random effects
Parameter	Mean (Raw metric)	Mean (Within-level standardized)	Variance
iM	6.51* [6.47, 6.54]	-	0.048* [0.038, 0.060]
ϕ	0.08* [0.06, 0.11]	0.08* [0.06, 0.11]	0.012* [0.006, 0.020]
trial	-0.01* [-0.01, -0.01]	-0.06* [-0.08, -0.04]	0.001* [0.001, 0.001]
	Covariance	Correlation	
iM ↔ ϕ	-0.01 [-0.01, 0.01]	-.08 [-.37, .22]	-
iM↔ trial	-0.01* [-0.01, -0.01]	-.62* [-.71, -.50]	-
ϕ ↔ trial	0.01 [-0.01, 0.01]	.06 [-.18, .29]	-

Note: *95% CI does not include 0.

To assess the global costs of having to perform the PM task on top of the OT task, we then conducted paired sample t-test for iM, iSD, ϕ, and OT accuracy. As reported in Table C2, participants were overall slower ($t(196) = 5.34$, $p < .001$), had larger fluctuations ($t(196) = 9.95$, $p < .001$), and greater inertia ($t(196) = 12.34$, $p < .001$) in OT RT in the PM block than in the OT only block. In addition, their OT accuracy also decreased from the OT only to the PM block ($t(196) = -10.52$, $p < .001$). The effect size for the difference between the two blocks was calculated using Cohen's d, indicating that these effects were large for ϕ (Cohen's $d = 0.91$), medium for iSD (Cohen's $d = 0.71$) and OT (Cohen's $d = -0.75$), and small for iM (Cohen's $d = 0.38$). These results confirm our expectation that participants have to recruit further attentional processes to be able to carry out the PM task on top of the OT. They further indicate that although there is no direct cost of checking the clock to OT accuracy, general performance in the OT still decreases when PM task requirements are added to the OT. Whether this cost comes from having to maintain the PM intention throughout the block or from increased iSD in the TBPM block remains to be clarified.

Table C2. Mean Costs for iM, iSD, ϕ, and OT, and Corresponding Paired Sample T-tests

Parameters	OT Only block		TBPM block		Mean difference (cost)	*t*(196)	*p*	Cohen's *d*
	M	*SD*	*M*	*SD*				
iM	6.51	0.20	6.57	0.13	0.06	5.34	< .001	0.38
iSD	0.24	0.05	0.28	0.05	0.04	9.95	< .001	0.71
ϕ	0.08	0.06	0.16	0.06	0.07	12.75	< .001	0.91
OT	91.45	8.95	87.86	5.56	-4.02	-10.52	< .001	-0.75

Note. OT only block data were missing for one participant.